Natural Frequency for Soil Deposit—NAFR

The program NAFR (**Na**tural **Fr**equency of Soil Deposit) is a subroutine subprogram that looks for peaks in the amplified spectrum in order of decreasing frequency and finds the natural frequencies from the first order to the specified order or to the highest order that can be found..

NAFR (**Na**tural **Fr**equency of Soil Deposit）

【Purpose】

To compute the natural frequency of the ground from the peak position of the amplified spectrum.

【Usage】

(1) How to connect

```
CALL NAFR (N, DF, FREQ, AMP, ND1, NMODE, F, ND2)
```

Argument	Type	Parameter in calling program	Return Parameter
N	I	Total number of amplified spectrum	Unchanged
DF	R	Frequency increment of amplified spectrum (unit : Hz)	Unchanged
FREQ	R 1-D array (ND1)	Frequency string of amplified spectrum (unit :Hz)	Unchanged
AMP	R 1-D array (ND1)	Amplified spectrum data	Unchanged
ND1	I	Dimension size of FREQ and AMP in calling program	Unchanged
NMODE	I	No need to input here	Highest order of natural frequency obtained
F	R 1-D array (ND2)	No need to input here	Natural Frequencies (unit : Hz)
ND2	I	Dimension size of F in calling program	Unchanged

(2) Necessary subroutines and function subprograms

None

(3) Remarks

It is not able to find natural frequencies above the *ND2* order. Therefore, *ND2* can be used to specify the highest order of the natural frequency to be determined.

【Calculation Method】

The point where the data of the amplification spectrum shows the peak and the two points before and after it are taken as a total of three points, and the frequency of the point where the quadratic curve passing through these three points shows the maximum value is defined as the natural frequency. In other words, if the number of data in the amplified spectrum is *N*, and the three successive data A_{k-1}, A_k, and A_{k+1} $(k = 2, 3, \cdots, N-1)$ are

$$A_{k-1} \leq A_k \geq A_{k+1}$$

then the peak lies between A_{k-1} and A_{k+1}. Then, when the frequencies of the three points A_{k-1}, A_k , and A_{k+1} are $f_{k-1} = f_k - \Delta f$, f_k, $f_{k+1} = f_k + \Delta f$, respectively, the equation of the quadratic curve passing through the three points is

$$A = af^2 + bf + c \qquad \text{(a)}$$

where

$$a = \frac{A_{k-1} - 2A_k + A_{k+1}}{2(\Delta f)^2} \ ,$$

$$b = -\frac{(A_{k-1} - A_{k+1})\Delta f + 2f_k(A_{k-1} - 2A_k + A_{k+1})}{2(\Delta f)^2} \ .$$

Therefore, by substituting *a* and *b* in the above equation for the value of $f = -b/2a$, which is obtained from the condition $dA/df = 2af + b = 0$ in equation (a), the frequency that gives the peak is determined by the following equation.

$$f = f_k + \frac{A_{k-1} - A_{k+1}}{A_{k-1} - 2A_k + A_{k+1}} \cdot \frac{\Delta f}{2}$$

To obtain the amplified spectrum, you can use the subroutine **FESP** described separately.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *          NAFR   1
C     SUBROUTINE FOR NATURAL FREQUENCIES OF LAYERED SOIL                  NAFR   2
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *          NAFR   3
C                                                                         NAFR   4
C                                      CODED BY Y.OHSAKI                  NAFR   5
C                                                                         NAFR   6
      SUBROUTINE NAFR(N,DF,FREQ,AMP,ND1,NMODE,F,ND2)                      NAFR   7
C                                                                         NAFR   8
      DIMENSION FREQ(ND1),AMP(ND1),F(ND2)                                 NAFR   9
C                                                                         NAFR  10
      NMODE=0                                                             NAFR  11
      DO 110 K=2,N-1                                                      NAFR  12
      IF(NMODE.EQ.ND2) RETURN                                             NAFR  13
```

```
      IF(AMP(K-1).GT.AMP(K).OR.AMP(K).LT.AMP(K+1))  GO TO 110
      NMODE=NMODE+1
      F(NMODE)=(AMP(K-1)-AMP(K+1))/(AMP(K-1)-2.*AMP(K)+AMP(K+1))
     *         *DF/2.+FREQ(K)
  110 CONTINUE
      RETURN
      END
```

【Example】

Calculate the amplification spectrum of the ground from the first to the third layer of the four-layer model ground with the characteristic values shown in the table below (the damping coefficients are $\alpha = 2.0$ and $\beta = 0.02$ for each layer), and calculate the natural frequency and natural period using **NAFR** based on this spectrum..

Layer No.	Upper depth (m)	Layer thickness (m)	Unit volume weight (tf/m^3)	Elastic shear modulus (tf/m^2)
1	0.00	3.80	1.50	1200
2	3.80	3.20	1.67	2900
3	7.00	3.90	1.85	5700
4	10.90	-	1.95	50000

```
C
      COMPLEX    Z(1000)
      DIMENSION  TH(4),UW(4),G(4),ALPHA(4),BETA(4),FREQ(1000),
     *           AMP(1000),F(5)
      DATA       L/4/,TH/3.8,3.2,3.9,0.0/,UW/1.50,1.67,1.85,1.95/,
     *           G/1200.,2900.,5700.,50000/,ALPHA/4*2.0/,BETA/4*0.02/,
     *           LOBJ/1/,LREF/4/,N/1000/,DF/0.02/
C
      CALL  FESP(L,TH,UW,G,ALPHA,BETA,4,LOBJ,LREF,DF,N,Z,1000,0,0)
      DO  110  K=1,N
      FREQ(K)=REAL(K-1)*DF
      AMP(K)  =CABS(Z(K))
 110  CONTINUE
      CALL  NAFR(N,DF,FREQ,AMP,1000,NMODE,F,5)
      WRITE(6,601)  LOBJ,LREF
      DO  120  J=1,NMODE
      T=1.0/F(J)
      WRITE(6,602)  J,F(J),T
 120  CONTINUE
      STOP
C
  601 FORMAT('LAYER',I2'/ LAYER',I2//T6,'MODE',TR4,'FREQ.(HZ)',
     *                  TR3,'PERIOD(SEC)' )
  602 FORMAT(T6,I3,F13.3,F12.3)
      END
```

Output:

LAYER 1/ LAYER 4

MODE	FREQ.(HZ)	PERIOD(SEC)
1	3.533	0.283
2	8.219	0.122
3	14.037	0.071
4	18.849	0.053